

PEEK AND POKE
CAMPER COPY

***** PEEK AT THE LEFT MARGIN SETTING *****

PRINT PEEK(82)

***** POKE THE LEFT MARGIN WITH A NEW VALUE *****

5, 10, 20, 39

***** POKING THE LEFT MARGIN USING A VARIABLE *****

```
10 REM **    POKING THE LEFT MARGIN
20 REM *
30 PRINT
40 FOR LEFTMARGIN = 0 TO 39 STEP 5
50 POKE 82,COUNT
60 PRINT
70 PRINT "LEFT MARGIN "
80 NEXT LEFTMARGIN
```

***** PEEK AT THE RIGHT MARGIN *****

PRINT PEEK(83)

***** POKE THE RIGHT MARGIN *****

POKE 83,20

PEEK AND POKE
CAMPER COPY CONTINUED

******* POKING THE RIGHT MARGIN WITH A VARIABLE *******

```
100 REM **      DECREMENT RIGHT MARGIN
110 REM *
120 PRINT
130 FOR RIGHTMARGIN = 39 TO 0 STEP -1
140 POKE 83,RIGHTMARGIN
150 PRINT "RIGHT MARGIN";
160 FOR DELAY = 1 TO 50: NEXT DELAY
170 NEXT RIGHTMARGIN
```

******* INVERTED PRINT *******

POKE 755,6

POKE 755,2

******* REFERENCES WITH MORE IDEAS FOR MEMORY LOCATIONS
TO PEEK AND POKE *******

Atari Connection, Summer,1983: pp 31-32
Inside Atari BASIC: pp 132-139
Master Memory Map

READ, DATA, AND RESTORE

CAMPER COPY CONTINUED

```

100 REM *      ART SHOW
105 REM *
110 MENU=900:REM MENU LINE NUMBER
115 GRAPHICS 7:COLOR 3
120 REM *
125 REM *****      MAIN LOOP      *****
130 GOSUB MENU
140 INPUT RESPONSE
150 IF RESPONSE<1 OR RESPONSE>4 THEN 140
160 IF RESPONSE=1 THEN RESTORE 500
170 IF RESPONSE=2 THEN RESTORE 600
180 IF RESPONSE=3 THEN RESTORE 700
190 IF RESPONSE=4 THEN RESTORE 800
200 REM *
210 REM *****      DRAW ROUTINE      *****
220 REM *
230 READ X,Y
240 PLOT X,Y:REM PICTURE START POINT
250 READ X,Y:REM GET DRAWTO DATA
260 IF X=-1 THEN 130:REM THE FLAG?
270 DRAWTO X,Y
280 GOTO 250:REM GET MORE DATA
290 REM *
500 REM *****      MOUNTAIN      *****
510 REM *
520 DATA 0,26,12,20,20,23,30,18,35,12,42,13,45,10,58,6
530 DATA 62,3,70,1,82,3,90,8,102,20,112,26,120,23
540 DATA 130,38,135,36,150,43,-1,-1
550 REM *
600 REM *****      BARN      *****610 REM *
610 REM *
620 DATA 43,50,43,46,47,46,47,50,40,50
630 DATA 40,44,45,41,50,44,50,50,40,50,-1,-1
640 REM *
700 REM *****      STAR      *****
710 REM *
720 DATA 128,10,127,11,126,11,127,12,126,13,127,13,128,14,129,13
730 DATA 130,13,129,12,130,11,129,11,128,11,128,14,-1,-1
740 REM *
800 REM *****      HORSE      *****
810 REM *
820 DATA 52,47,54,46,54,45,54,50,54,48,57,48,58,47,57,48,57,50,-1,-1
900 REM *
910 REM *****      MENU      *****
920 REM *
930 PRINT
940 PRINT "1.  MOUNTAIN      3.  STAR"
950 PRINT "2.  BARN        4.  HORSE"
960 PRINT "WHICH PICTURE (1, 2, 3, OR 4) ";
970 RETURN

```

READ, DATA, AND RESTORE
CAMPER COPY CONTINUED

***** Trapped DATA *****

```
100 REM *      TRAPPED DATA
110 REM *
120 DIM PERSON$(20)
130 PRINT
140 READ PERSON$
150 TRAP 210
160 READ AGE
170 PRINT PERSON$;" IS ";AGE;" YEARS OLD."
180 GOTO 140
190 DATA MARGIE,47,JOHN,50,BETH,12
200 DATA
210 END
```

***** Music *****

```
100 REM *      MUSIC
110 REM *
120 READ PITCH,DISTORT,LOUD,TIME
130 IF TIME = -1 THEN GOTO 210
140 SOUND 0,PITCH,DISTORT,LOUD
150 FOR DELAY = 1 TO TIME : NEXT DELAY
160 GOTO 120
170 DATA 121,10,10,40,91,10,10,37
180 DATA 0,0,0,3,91,10,10,40,108,10,10,28
190 DATA 0,0,0,2,108,10,10,10,91,10,10,30
200 DATA 108,10,10,10,121,10,10,80,0,0,0,0,-1
210 END
```

READ, DATA, AND RESTORE
CAMFER COPY

***** Read DATA Demo *****

```
10 REM * READ DATA DEMO
20 REM *
30 DIM DAY$(20)
40 NUMOFDAYS = 7
50 PRINT
60 FOR COUNTER = 1 TO NUMOFDAYS
70 READ DAY$
80 PRINT DAY$
90 NEXT COUNTER
100 DATA SUNDAY, MONDAY, TUESDAY
110 DATA WEDNESDAY, THURSDAY, FRIDAY
120 DATA SATURDAY
```

***** READ Two String Variables *****

```
100 REM * TWO STRING VARIABLES
110 REM *
120 DIM DAY$(15), ORDER$(10)
130 PRINT
140 READ DAY$
150 IF DAY$="FINISH" THEN GOTO 220
160 READ ORDER$
170 PRINT DAY$;" IS THE ";ORDER$;" OF THE WEEK."
180 GOTO 140
190 DATA SUNDAY,FIRST,MONDAY,SECOND,TUESDAY,THIRD
200 DATA WEDNESDAY,FOURTH,THURSDAY,FIFTH,FRIDAY,SIXTH
210 DATA SATURDAY,SEVENTH,FINISH
220 END
```

190 DATA SUNDAY,FIRST,MONDAY,SECOND,TUESDAY,THIRD,FINISH

190 DATA SUNDAY,FIRST,MONDAY,SECOND,TUESDAY,THIRD, FINISH



PROGRAMMING CHALLENGES USING READ, DATA, AND RESTORE

1. Write a program that lists the necessary values for a song in DATA statements. Use the RESTORE command to repeat the chorus of the song in between the verses.
2. Add pictures to the ART SHOW program.
3. Set up a PLOT subroutine like the DRAWTO subroutine in the ART SHOW program. Add a plotted image to the list of pictures one can put on the screen. For example, offer to draw stars and PLOT tiny dots in the sky using READ and DATA statements.
4. Allow the user to draw the shapes anywhere on the screen. For example, when a person selects the barn, ask for the coordinates of the cabin's location on the screen. Be sure to tell the user the range of possible coordinates when asking for INPUT. Then add the person's coordinates to the DATA coordinates in the DRAWTO statement (DRAWTO X+XCOORD,Y+YCOORD). This way the person can put as many barns on the screen as he or she wants as well as put them anywhere in the picture.
5. Draw the shapes in different colors. Store the color of the shape as the first element of the shape data. READ the value into the COLOR instruction before drawing the image.
6. Experiment with combining different graphic modes, colors, sounds, etc. into a little show using the READ, DATA, and RESTORE statements.